```
(*Problem I .6 .13*)

Apply[Clear, Names["Global`*"]];
Off[General::spell];
Off[General::spell1];

(*Input data*)

AB = 0.25;
BC = 0.94;
DE = 0.38;
CD = 0.38;
EF = 0.7;
CE = CD + DE;
La = .93;
Lb = .31;
Lc = .31;
phi = N[Pi] 120. / 180;

(*Position of joint A*)
xA = yA = 0;
rA = {xA, yA, 0};
Print["rA = ", rA, " m"];

(*Position of joint D*)
xD = -La;
yD = Lb;
rD = {xD, yD, 0};
Print["rD = ", rD, " m"];

(*Position of joint B*)
xB = AB Cos[phi];
yB = AB Sin[phi];
rB = {xB, yB, 0};
Print["rB = ", rB, " m"];

(*Position of joint C*)
eqn41 = (xCsol - xD)^2 + (yCsol - yD)^2 - CD^2 == 0;
eqn42 = (xCsol - xB)^2 + (yCsol - yB)^2 - BC^2 == 0;
solutionC = Solve[{eqn41, eqn42}, {xCsol, yCsol}];
(*Two solutions for C*)
xC1 = xCsol /. solutionC[[1]];
yC1 = yCsol /. solutionC[[1]];
xC2 = xCsol /. solutionC[[2]];
yC2 = yCsol /. solutionC[[2]];
(*Select the correct position for C*)
If [(yC1) < (yD) , xC = xC1; yC = yC1, xC = xC2; yC = yC2];
rC = {xC, yC, 0};
Print["rC = ", rC, " m"];

(*Position of Joint E*)
phi3 = ArcTan[(yD - yC) / (xD - xC)];
```

```
xE = xC + ((CD + DE) (Cos[phi3]));
yE = yC + ((CD + DE) (Sin[phi3]));
rE = {xE, yE, 0};
Print["rE = ", rE, " m"];

(*Position of Joint F*)
yF = Lb + Lc;
xF = xE + Sqrt[EF^2 - (yF - yE)^2];
rF = {xF, yF, 0};
Print["rF = ", rF, " m"];

(*Graph of the mechanism*)

markers = Table[{
   Point[{xA, yA}],
   Point[{xB, yB}],
   Point[{xC, yC}],
   Point[{xD, yD}],
   Point[{xE, yE}],
   Point[{xF, yF}]
  }];

name = Table[{
    Text["A", {xA, yA}, {-1, 1}],
    Text["B", {xB, yB}, {0, -1}],
    Text["C", {xC, yC}, {0, -1}],
    Text["D", {xD, yD}, {0, -1}],
    Text["E", {xE, yE}, {-1, 1}],
    Text["F", {xF, yF}, {0, -1}]
   }];

graph = Graphics[
  {{RGBColor[1, 0, 0],
    Line[{{xA, yA}, {xB, yB}}]},
   {RGBColor[0, 1, 0],
    Line[{{xB, yB}, {xC, yC}}]},
   {RGBColor[0, 0, 1],
    Line[{{xC, yC}, {xE, yE}}]},
   {RGBColor[1, 0, 1],
    Line[{{xE, yE}, {xF, yF}}]},
   {RGBColor[1, 1, 1],
    markers},
   {name}}];

Show[Graphics[graph], PlotRange → {{-1.2, .1}, {-.1, .8}},
  Frame → True, AxesOrigin → {xA, yA}, FrameLabel → {"x", "y"},
  Axes → {True, True}, AspectRatio → Automatic];

n = 1500.; (*rpm*)
ω = n * N[Pi] / 30; (*rad/s*)

Print["0-Ar-1-Br-2-Cr-3-Dr-0"];
```

```
ω10 = {0, 0, ω};
ω21v = {0, 0, omega21};
ω32v = {0, 0, omega32};
ω03v = {0, 0, omega03};

eqIkv = (ω10 + ω21v + ω32v + ω03v)[[3]] == 0;
eqIijv = Cross[rB, ω21v] + Cross[rC, ω32v] + Cross[rD, ω03v];
eqIiv = eqIijv[[1]] == 0;
eqIjv = eqIijv[[2]] == 0;
solIvel =
 Solve[{eqIkv, eqIiv, eqIjv}, {omega21, omega32, omega03}];

omega21s = omega21 /. solIvel[[1]];
omega32s = omega32 /. solIvel[[1]];
omega03s = omega03 /. solIvel[[1]];

ω21 = {0, 0, omega21s};
ω32 = {0, 0, omega32s};
ω03 = {0, 0, omega03s};

ω20 = ω10 + ω21;
ω30 = -ω03;

Print["ω20 = ", ω20, " rad/s"];
Print["ω30 = ", ω30, " rad/s"];

vB = Cross[ω10, rB];
vC = vB + Cross[ω20, rC - rB];
vE = Cross[ω30, rE - rD];

Print["vB = ", vB, "  m/s"];
Print["vC = ", vC, "  m/s"];
Print["vE = ", vE, "  m/s"];

(*accelerations*)

α10 = D[ω10, t];
α21v = {0, 0, alpha21};
α32v = {0, 0, alpha32};
α03v = {0, 0, alpha03};

eqIka = (α10 + α21v + α32v + α03v)[[3]] == 0;
eqIija = Cross[rB, α21v] + Cross[rC, α32v] + Cross[rD, α03v] -
  ω10.ω10 rB - ω20.ω20 (rC - rB) - ω30.ω30 (rD - rC);
eqIia = eqIija[[1]] == 0;
eqIja = eqIija[[2]] == 0;
solIacc =
 Solve[{eqIka, eqIia, eqIja}, {alpha21, alpha32, alpha03}];

alpha21s = alpha21 /. solIacc[[1]];
alpha32s = alpha32 /. solIacc[[1]];
alpha03s = alpha03 /. solIacc[[1]];
```

```
α21 = {0, 0, alpha21s};
α32 = {0, 0, alpha32s};
α03 = {0, 0, alpha03s};

α20 = α10 + α21;
α30 = -α03;
Print["α20 = ", α20, " rad/s^2"];
Print["α30 = ", α30, " rad/s^2"];

aB = Cross[α10, rB] - ω10.ω10 rB;
aC = aB + Cross[α20, rC - rB] - ω20.ω20 (rC - rB);
aE = Cross[α30, rE - rD] - ω30.ω30 (rE - rD);

Print["aB = ", aB, "  m/s^2"];
Print["aC = ", aC, "  m/s^2"];
Print["aE = ", aE, "  m/s^2"];

Print[" "];

Print["0-Dr-3-Er-4-Fr-5-Ft-0"];

ω43v = {0, 0, omega43};
ω54v = {0, 0, omega54};
vF05v = {vF05, 0, 0};

eqIIkv = (ω30 + ω43v + ω54v)[[3]] == 0;
eqIIijv = Cross[rD, ω30] + Cross[rE, ω43v] + Cross[rF, ω54v] + vF05v;
eqIIiv = eqIIijv[[1]] == 0;
eqIIjv = eqIIijv[[2]] == 0;
solIvelI =
 Solve[{eqIIkv, eqIIiv, eqIIjv}, {omega43, omega54, vF05}];

omega43s = omega43 /. solIvelI[[1]];
omega54s = omega54 /. solIvelI[[1]];
vF05s = vF05 /. solIvelI[[1]];

ω43 = {0, 0, omega43s};
ω54 = {0, 0, omega54s};
v05 = {vF05s, 0, 0};

ω40 = ω30 + ω43;

Print["ω40 = ", ω40, " rad/s"];

vF = vE + Cross[ω40, rF - rE];

Print["vF = ", Chop[vF], "  m/s"];

(*accelerations*)

α43v = {0, 0, alpha43};
α54v = {0, 0, alpha54};
```

```
aF05v = {aF05, 0, 0};

eqIIka = (α30 + α43v + α54v)[[3]] == 0;
eqIIija = Cross[rD, α30] + Cross[rE, α43v] +
  Cross[rF, α54v] + aF05v - ω30.ω30 (rE - rD) - ω40.ω40 (rF - rE);
eqIIia = eqIIija[[1]] == 0;
eqIIja = eqIIija[[2]] == 0;
solIIa = Solve[{eqIIka, eqIIia, eqIIja}, {alpha43, alpha54, aF05}];

alpha43s = alpha43 /. solIIa[[1]];
alpha54s = alpha54 /. solIIa[[1]];
aF05s = aF05 /. solIIa[[1]];

α43 = {0, 0, alpha43s};
α54 = {0, 0, alpha54s};
a05 = {aF05s, 0, 0};

α40 = α30 + α43;

Print["α40 = ", α40, " rad/s^2"];

aF = aE + Cross[α40, rF - rE] - ω40.ω40 (rF - rE);

Print["aF = ", Chop[aF], "  m/s^2"];

Print["--------------------------------"];

α2 = α20;
α3 = α30;
α4 = α40;
α5 = {0, 0, 0};
Fext = -Sign[vF] {2000., 0, 0};
Print["Fext = ", Fext, "  N"];

h = 0.01;
d = 0.001;
hSlider = 0.02;
wSlider = 0.05;
ρ = 8000.;
g = 9.807;

(*Link 1*)
m1 = ρ AB h d;
rC1 = rB / 2;
aC1 = aB / 2;
Fin1 = -m1 aC1;
G1 = {0, -m1 g, 0};
F1 = (Fin1 + G1);
IC1 = m1 (AB^2 + h^2) / 12;
M1 = Min1 = -IC1 α10;
Print["rC1 = ", rC1, "  m"];
Print["aC1 = ", aC1, "  m/s^2"];
Print["m1 = ", m1, "  kg"];
Print["IC1 = ", IC1, "  kg m^2"];
```

```
Print["Fin1 = -m1 aC1 = ", Fin1, "  N "];
Print["G1 = -m1 g = ", G1, "  N "];
Print["F1 = -m1 aC1 + G1 = ", F1, "  N "];
Print["M1 = Min1 = -IC1 α1 = ", {0, 0, 0}, "  N m"];

(*Link 2*)
m2 = ρ BC h d;
rC2 = (rB + rC) / 2;
aC2 = (aB + aC) / 2;
Fin2 = -m2 aC2;
G2 = {0, -m2 g, 0};
F2 = (Fin2 + G2);
IC2 = m2 (BC^2 + h^2) / 12;
M2 = Min2 = -IC2 α2;
Print["rC2 = ", rC2, "  m"];
Print["aC2 = ", aC2, "  m/s^2"];
Print["m2 = ", m2, "  kg"];
Print["IC2 = ", IC2, "  kg m^2"];
Print["Fin2 = -m2 aC2 = ", Fin2, "  N "];
Print["G2 = -m2 g = ", G2, "  N "];
Print["F2 = -m2 aC2 + G2 = ", F2, "  N "];
Print["M2 = Min2 = -IC2 α2 = ", M2, "  N m"];

(*Link 3*)
m3 = ρ CE h d;
rC3 = (rC + rE) / 2;
aC3 = (aC + aE) / 2;
Fin3 = -m3 aC3;
G3 = {0, -m3 g, 0};
F3 = Chop[(Fin3 + G3)];
IC3 = m3 (CE^2 + h^2) / 12;
M3 = Min3 = -IC3 α3;
Print["rC3 = ", rC3, "  m"];
Print["aC3 = ", aC3, "  m/s^2"];
Print["m3 = ", m3, "  kg"];
Print["IC3 = ", IC3, "  kg m^2"];
Print["Fin3 = -m3 aC3 = ", Fin3, "  N "];
Print["G3 = -m3 g = ", G3, "  N "];
Print["F3 = -m3 aC3 + G3 = ", F3, "  N "];
Print["M3 = Min3 = -IC3 α3 = ", M3, "  N m"];

(*Link 4*)
m4 = ρ EF h d;
rC4 = (rE + rF) / 2;
aC4 = (aE + aF) / 2;
Fin4 = -m4 aC4;
G4 = {0, -m4 g, 0};
F4 = (Fin4 + G4);
IC4 = m4 (EF^2 + h^2) / 12;
M4 = Min4 = -IC4 α4;
Print["rC4 = ", rC4, "  m"];
Print["aC4 = ", aC4, "  m/s^2"];
Print["m4 = ", m4, "  kg"];
Print["IC4 = ", IC4, "  kg m^2"];
Print["Fin4 = -m4 aC4 = ", Fin4, "  N "];
```

```
Print["G4 = -m4 g = ", G4, "  N "];
Print["F4 = -m4 aC4 + G4 = ", F4, "  N "];
Print["M4 = Min4 = -IC4 α4 = ", M4, "  N m"];

(*Link 5*)
m5 = ρ hSlider wSlider d;
rC5 = rF;
aC5 = aF;
Fin5 = -m5 aC5;
G5 = {0, -m5 g, 0};
F5 = (Fin5 + G5);
IC5 = m5 (hSlider^2 + wSlider^2) / 12;
M5 = Min5 = -IC5 α5;
Print["rC5 = ", rC5, "  m"];
Print["aC5 = ", aC5, "  m/s^2"];
Print["m5 = ", m5, "  kg"];
Print["IC5 = ", IC5, "  kg m^2"];
Print["Fin5 = -m5 aC5 = ", Fin5, "  N "];
Print["G5 = -m5 g = ", G5, "  N "];
Print["F5 = -m5 aC5 + G5 = ", F5, "  N "];
Print["M5 = Min5 = -IC5 α5 = ", M5, "  N m"];

Print[" "];
Print[" 4&5 "];
F05 = {0, F05y, 0};
F34 = {F34x, F34y, 0};
eqF45 = (F5 + F4 + F05 + F34 + Fext);
eqMF4 = (Cross[rE - rF, F34] + Cross[rC4 - rF, F4] + M4)[[3]];
sol45 = Solve[
   {eqF45[[1]] == 0, eqF45[[2]] == 0, eqMF4 == 0}, {F05y, F34x, F34y}];
F05s = F05 /. sol45[[1]];
F34s = F34 /. sol45[[1]];
Print["F05 = ", F05s, "  N"];
Print["F34 = ", F34s, "  N"];
F54 = F05s + F5 + Fext;
Print["F54 = ", F54, "  N"];

Print[" "];
Print[" 2&3 "];
F12 = {F12x, F12y, 0};
F03 = {F03x, F03y, 0};
eqF23 = (F2 + F3 + F03 + F12 - F34s);
eqMC3 = (Cross[rD - rC, F03] +
    Cross[rC3 - rC, F3] + Cross[rE - rC, -F34s] + M3)[[3]];
eqMC2 = (Cross[rB - rC, F12] + Cross[rC2 - rC, F2] + M2)[[3]];
sol23 = Solve[{eqF23[[1]] == 0, eqF23[[2]] == 0, eqMC3 == 0, eqMC2 == 0},
   {F03x, F03y, F12x, F12y}];
F03s = F03 /. sol23[[1]];
F12s = F12 /. sol23[[1]];
Print["F03 = ", F03s, "  N"];
Print["F12 = ", F12s, "  N"];
F23 = F12s + F2;
Print["F23 = ", F23, "  N"];
```

```
Print["  "]
(*Link 1*)
F01 = -F1 + F12s;
Print["F01 = ", F01, "  N"];
Mm = -Cross[rB, -F12s] - Cross[rC1, F1] - M1;
Print["Meq = ", Mm, "  N m"];
```

rA = {0, 0, 0} m

rD = {-0.93, 0.31, 0} m

rB = {-0.125, 0.216506, 0} m

rC = {-1.02396, -0.0581993, 0} m

rE = {-0.836036, 0.678199, 0} m

rF = {-0.138459, 0.62, 0} m



0-Ar-1-Br-2-Cr-3-Dr-0

ω20 = {0, 0, -34.1602} rad/s

ω30 = {0, 0, -117.851} rad/s

vB = {-34.0087, -19.635, 0.} m/s

vC = {-43.3927, 11.0738, 0.} m/s

vE = {43.3927, -11.0738, 0.} m/s

α20 = {0, 0, -11357.4} rad/s^2

α30 = {0, 0, -792.294} rad/s^2

aB = {3084.25, -5342.08, 0.} m/s^2

aC = {1013.34, 5188.33, 0.} m/s^2

aE = {-1013.34, -5188.33, 0.} m/s^2

0-Dr-3-Er-4-Fr-5-Ft-0

$\omega 40$ = {0, 0, 15.8747} rad/s

vF = {44.3166, 0, 0} m/s

$\alpha 40$ = {0, 0, 7416.63} rad/s^2

aF = {-757.49, 0, 0} m/s^2

Fext = {-2000., 0, 0} N

rC1 = {-0.0625, 0.108253, 0} m

aC1 = {1542.13, -2671.04, 0.} m/s^2

m1 = 0.02 kg

IC1 = 0.000104333 kg m^2

Fin1 = -m1 aC1 = {-30.8425, 53.4208, 0.} N

G1 = -m1 g = {0, -0.19614, 0} N

F1 = -m1 aC1 + G1 = {-30.8425, 53.2247, 0.} N

M1 = Min1 = -IC1 $\alpha 1$ = {0, 0, 0} N m

rC2 = {-0.574482, 0.0791535, 0} m

aC2 = {2048.8, -76.8741, 0.} m/s^2

m2 = 0.0752 kg

IC2 = 0.00553785 kg m^2

Fin2 = -m2 aC2 = {-154.069, 5.78093, 0.} N

G2 = -m2 g = {0, -0.737486, 0} N

F2 = -m2 aC2 + G2 = {-154.069, 5.04345, 0.} N

M2 = Min2 = -IC2 $\alpha 2$ = {0, 0, 62.8953} N m

rC3 = {-0.93, 0.31, 0} m

aC3 = $\{4.54747\times10^{-13}, -2.72848\times10^{-12}, 0.\}$ m/s^2

m3 = 0.0608 kg

IC3 = 0.00292701 kg m^2

Fin3 = -m3 aC3 = $\{-2.76486\times10^{-14}, 1.65892\times10^{-13}, 0.\}$ N

G3 = -m3 g = {0, -0.596266, 0} N

F3 = -m3 aC3 + G3 = {0, -0.596266, 0} N

M3 = Min3 = -IC3 $\alpha 3$ = {0, 0, 2.31906} N m

rC4 = {-0.487247, 0.6491, 0} m

aC4 = {-885.414, -2594.17, 0.} m/s^2

m4 = 0.056 kg

IC4 = 0.00228713 kg m^2

Fin4 = -m4 aC4 = {49.5832, 145.273, 0.} N

G4 = -m4 g = {0, -0.549192, 0} N

F4 = -m4 aC4 + G4 = {49.5832, 144.724, 0.} N

M4 = Min4 = -IC4 α4 = {0, 0, -16.9628} N m

rC5 = {-0.138459, 0.62, 0} m

aC5 = $\{-757.49, 8.5798\times10^{-13}, 0.\}$ m/s^2

m5 = 0.008 kg

IC5 = 1.93333×10^{-6} kg m^2

Fin5 = -m5 aC5 = $\{6.05992, -6.86384\times10^{-15}, 0.\}$ N

G5 = -m5 g = {0, -0.078456, 0} N

F5 = -m5 aC5 + G5 = {6.05992, -0.078456, 0.} N

M5 = Min5 = -IC5 α5 = {0, 0, 0} N m

4&5

F05 = {0, 116.321, 0} N

F34 = {1944.36, -260.966, 0} N

F54 = {-1993.94, 116.242, 0.} N

2&3

F03 = {4144.38, 455.816, 0} N

F12 = {-2045.95, -721.229, 0} N

F23 = {-2200.02, -716.186, 0.} N

F01 = {-2015.11, -774.454, 0.} N

Meq = {0., 0., 533.104} N m